

Propensity Score-based Analysis of Short-Term Complications in Patients with Lumbar Discectomy in the ACS-NSQIP Database

Yubo Gao, Ph.D, Department of Orthopaedic Surgery & Rehabilitation, The University of Iowa



INTRODUCTION

- Lumbar discectomy can be done on outpatient or inpatient basis with varied benefits.
- There is no large study comparing the incidence of complications between inpatient and outpatient undergoing discectomy.

PURPOSE

- To compare the incidence of complications in patients undergoing discectomy between the inpatient and outpatient settings.
- To determine baseline 30-day complication rates, and
- To identify independent risk factors for complications.

MATERIALS & METHODS

- Patients were identified from the ACS NSQIP database from 2005 to 2010 using CPT code 63030. Total 4310 cases.

Explanatory/Control variables

- Patient demographics: age, sex, and race
- Preoperative variables: BMI, recent weight loss, diabetes, and *et al*
- Preoperative comorbidities: coronary artery disease, and *et al*.
- Operative variables: wound class, ASA class, and *et al*.

Outcomes

30-day complications after spine surgery include

- wound complications
- pulmonary complication
- hematologic complication
- cardiac complication
- renal complication
- neurologic complication
- And others

Statistical analysis

Unadjusted univariate comparison

- Proc ttest
- Proc freq

Propensity score_based model to reduce selection bias

- matching
- regression

SAS CODES

Propensity scores

```
proc logistic data=pgy2 descending;
  class sex new_race wtloss diabetes1 dialysis
        wndinf proper30 wndclas asaclas;
  model inout2=sex new_race wtloss diabetes1
        dialysis wndinf proper30
        wndclas asaclas age1 bmi
        prwbc prhct optime rbc/lackfit
        risklimits;
  output out=propensity_scores pred=prob_inout;
run;
quit;
```

Evaluation of common support

```
proc univariate data=propensity_scores noprint;
  class inout2;
  histogram prob_inout;
  inset N='N' (6.0)
  Mean='Mean' (4.2) STD='STD' (4.2)/font='Arial'
  pos=ne height=3;
run;
```

Dividing propensity score into 5 groups

```
proc rank groups=5 data=propensity_scores out=r;
  ranks rnks;
  var prob_inout;
run;
data a;
  set r;
  quintile=rnks+1;
Run;
```

Regression

```
proc logistic data=a descending;
  class inout2;
  model complication= inout2 quintile/lackfit risklimits;
run;
quit;

proc glm data=a;
  class inout2 ;
  model continuous_variables=inout2 quintile/ss3 solution;
  lsmeans inout2/adjust=tukey pdiff cl;
quit;
```

Matching by Hash

```
data outpatients0(rename=(prob_inout=pscoreT))
  inpatients0(rename=(prob_inout=pscoreC));
  set a;
  if prob_inout=. then delete;
  RandomNumber= ranuni(2011);
  if inout2=0 then output outpatients0; /*Treatment*/
  else if inout2=1 then output inpatients0; /*Control*/
run;
data outpatients0; /*Treatment*/
  set outpatients0;
  idT=_n_;
run;
data inpatients0; /*Control*/
  set inpatients0;
  idC=_n_;
run;
* Randomly sort the tables;
proc sort data= outpatients0 out= outpatients (keep=pscoreT
  idT);
  by RandomNumber;
run;
proc sort data= inpatients0 out= inpatients (keep=pscoreC
  idC);
  by RandomNumber;
run;
data Matched(keep= IdSelectedControl MatchedToTreatID);
  length pscoreC 8;
  length idC 8;
  if _N_ = 1 then do;
    declare hash h(dataset: "inpatients", ordered: 'no');
    declare hiter iter('h');
    h.defineKey('idC');
    h.defineData('pscoreC', 'idC');
    h.defineDone();
    call missing(idC, pscoreC);
  end;
  set outpatients; /*Treatment*/
  retain BestDistance 99;
  rc= iter.first();
  if (rc=0) then BestDistance= 99;
  do while (rc= 0);
    ScoreDistance= abs(pscoreT - pscoreC);
    if ScoreDistance < BestDistance then do;
      BestDistance= ScoreDistance;
      IdSelectedControl= idC;
```

SAS CODES-cont.

```
  MatchedToTreatID= idT;
end;
rc= iter.next();
if (rc≠ 0) then do;
  output;
  rc1= h.remove(key: IdSelectedControl);
end;
end;
run;
/*select those matched inpatients*/
proc sql noprint;
  create table Tinpatients as
  select a.*,b.*
  from matched a join inpatients b
  on a.IdSelectedControl=b.idC;
quit;
proc freq data=Tinpatients;
  tables idC;
run;
proc freq data=Tinpatients;
  tables IdSelectedControl;
run;
data analysis;
  set outpatients Tinpatients;
run;
data aa;
  set analysis;
  prob=pscoreT;
  if pscoreC ne . and pscoreT=. then prob=pscoreC;
Run;
/*after matching evaluation*/
proc univariate data=aa noprint;
  class inout2;
  histogram prob;
  inset N='N' (6.0)
  Mean='Mean' (4.2) STD='STD' (4.2)/font='Arial'
  pos=ne height=3;
run;
```

CONCLUSION

- Outpatients have lower peri-operative complication rates than inpatients
- Risk factors of short term complication: age, diabetes, presence of pre-operative wound infection, blood transfusion, operative time exceeding 150 minutes, and an inpatient hospital stay